Jak napisać grę logiczną

Na wstępie chciałbym zaznaczyć, że jest to artykuł przeznaczony dla osób średnio zaawansowanych, tj. takich, które poznały podstawy pisania kodu GML i chciałyby nauczyć się czegoś nowego. Ci bardziej doświadczeni będą się zapewne nudzić ale być może będą zainteresowany poznaniem mojego toku myślenia ;-)
Zakładam, że osoby czytające poznały strukturę tablic dwu-wymiarowych a także opanowały pisanie skryptów.

No to zaczynamy...

Napiszemy gierkę dla dwóch osób siedzących przy jednym komputerze. Gra polegać będzie na tym, żeby po jej zakończeniu na planszy zostało jak najwięcej pionków naszego koloru. Każdy z graczy zaczyna z 2 pionami swojego koloru. Aby zdobyć pionki przeciwnika należy je wziąć w kleszcze, tzn. dostawić swojego pionka tak, aby pionki przeciwnika znalazły się między naszymi. Jeśli zasady gry nie są jeszcze jasne, myślę, że wyjaśnią się podczas pisania kodu...

Zacznijmy od podstaw...

Żeby w ogóle mówić o grze trzeba zacząć od zbudowania planszy i ustawienia pionków. Ja ustaliłem sobie, że szerokość pola będzie 60 pixeli. Tworzymy zatem sprite o rozmiarach 60x60 z czarną ramką i szarym wypełnieniem. To będzie nasze pole. Następnie przygotujemy pionki. Pionek będzie oddzielnym obiektem. Żeby nie przeciążać niepotrzebnie systemu, nasz pionek będzie narysowany na polu, dzięki czemu po ustawieniu pionka będzie można usunąć niepotrzebne od tego momentu pole. Tworzymy więc kolejne pole (kopiujemy poprzednie ) oraz rysujemy na nim czarnego pionka. Ten sam sprite wykorzystujemy do stworzenia pionka białego(dodajemy klatkę). Mamy zatem dwu-klatkowy sprite z wizerunkiem czarnego ( w klatce zerowej ) oraz białego ( klatka 1 ) pionka na polu planszy. Na koniec tworzymy obiekty pole i wybieramy odpowiedni sprite oraz obiekt pion i jemu również wybieramy odpowiedni sprite.

Tworzymy plansze...

Każda gra jakoś się zaczyna. Nasza będzie zaczynać się zawsze tak samo ( plansza , punkty , aktualna tura , ustawienie pionków ) zatem stworzymy skrypt który ustawi wszystko w odpowiednim miejscu co znacznie ułatwi późniejszą pracę.

Plansza składa się z 64 pól ułożonych w kwadrat 8x8. Każdy z graczy na  początku 2 pionki swojego koloru ułożone na krzyż w centralnym obszarze planszy. Aby ułatwić sobie odwoływanie się do pól planszy, zamiast skakać po obiektach i wyszukiwać odpowiedniego obiektu na odpowiedniej pozycji, przedstawimy planszę w postaci tablicy dwuwymiarowej. Będzie to tablica 8x8 gdzie wartości wpisane w odpowiednie komórki będą stanowiły o tym jaki pionek aktualnie się tam znajduje. 

Ja przyjąłem że :
-1 oznacza pole puste

 0 oznacza czarnego pionka

 1 oznacza białego pionka
Można do tego celu przeznaczyć dowolne wartości, ale zauważmy, że w naszym przypadku łatwo odszukać analogię między reprezentacją liczbową pionka a wyświetlaną klatką. ( 0 – czarny , zerowa klatka sprite’a to czarny pionek itd.. ) Takie analogie ułatwiają znacznie pracę i należy je wykorzystywać kiedy to tylko możliwe. Przekonanie się o tym w dalszej części tego artykułu.

Nadeszła pora na stworzenie planszy i ustawienie na niej pionków. Ja stworzyłem skrypt o nazwie stworz_plansze i wpisałem do niego poniższy kod.

for (i=2;i<10;i+=1)
    for (j=1;j<9;j+=1)
    {
    if !(i==5&&(j==4||j==5) || i==6&&(j==4||j==5))
    {
    instance_create(i*60,j*60,pole);
    global.tablica[i,j]=-1;
    }}
global.tablica[5,4]=1;
global.tablica[6,4]=0; 
global.tablica[5,5]=0;
global.tablica[6,5]=1;   
for (i=5;i<7;i+=1)
    for (j=4;j<6;j+=1)
    {
    instance_create(i*60,j*60,pion);
    }

Skrypt wypełnia globalną tablicę dwu-wymiarową liczbami –1 wstawiając przy okazji obiekty pole w odpowiednie miejsce instance_create(i*60,j*60,pole);. Zauważmy jednak, że pomijamy przy tej czynności pola, które będą zajęte przez pionki, które są dane na początku gry. if !(i==5&&(j==4||j==5) || i==6&&(j==4||j==5))Na koniec wypełniamy centralną część planszy (tablicy) odpowiednimi wartościami ( 1 - czarny pionek, 0 - biały pionek ) oraz tworzymy pionki w tych właśnie miejscach instance_create(i*60,j*60,pion);

Tablica musi być znana ( wspólna ) wszystkim obiektom w grze dlatego należy zadeklarować ją globalnie. Zapewne zauważyliście, że tablica nie zaczyna się od komórki [0,0] tylko [2,1]. Zrobiłem tak dlatego, że pola wstawiane są za pomocą skryptu wykorzystującego współrzędne komórki tablicy  ( i*60,j*60 ). Gdybym zaczął wypełniać tablicę od 0 nasza plansza również zaczynałaby się od miejsca 0,0 ( 0*60 = 0 ;-) ) czyli lewego górnego rogu ekranu. Nie jest to zbyt efektowne i dlatego zrobiłem to co zrobiłem.

Skrypt należy wykonać tuż po utworzeniu room’u. Do tego celu doskonale nadaje się dział „creation code”, który znajduje się w każdnym pokoju w zakładce „settings”. Wpisujemy tam stworz_plansze();

Na koniec, należy jeszcze zadbać o to aby pionek postawiony w danym miejscu na planszy „wiedział” jakiego jest koloru na podstawie miejsca w jakim się znajduje. W tym celu tworzymy skrypt o nazwie ustal_kolor:

image_single=global.tablica[x/60,y/60];

Możemy tak zrobić, gdyż każdy obiekt ma dostęp do tablicy globalnej. Teraz widać czemu liczby w tablicy odpowiadają klatkom wyświetlanych spritów. Bez żadnych przekształceń możemy w łatwy sposób nakazać aby numer wyświetlanej klatki odpowiadał liczbie na odpowiednim miejscu w tablicy.

Skrypt wywołujemy w dziale CREATE dla pionka.

Zaczynamy gre
Kiedy mamy już gotową planszę a na niej pionki w odpowiednim kolorze, możemy przystąpić do definiowania zasad rozgrywki. Główna reguła to ta, opisująca zasady na jakich wstawiamy nowy pionek i od niej właśnie zaczniemy. Gra jest na dwóch graczy, zatem należy ustalić czyja jest aktualnie tura. Wiadomość tak musi być przekazana na samym początku rozgrywki wraz z definicją planszy. Dopiszmy zatem do działu „creation code” linijkę, która załatwi tą sprawę.

global.tura=1;

Dlaczego 1 ? Założę się, że nie pamiętacie już jak pisałem o analogiach. Otóż liczbę oznaczającą aktualną turę, możemy również wykorzystać do wielu rzeczy podczas rozgrywki. Żeby wam to ułatwić powiem np. że liczba 1 może oznaczać ze ruch ma gracz, którego pionki wyświetlają 1 klatkę animacji... tak białe. Dzięki temu nie musimy się już martwić o interpretację oznaczeń zmiennej global.tura.

Przejdźmy zatem do czegoś trudniejszego. Zastanówmy się jak będzie przebiegał każdy ruch gracza. Żeby nie łapać się za wszystko naraz przygotujemy sobie punktowy przebieg całej rozgrywki.

1) Kiedy zaczyna się tura ( ale nie tura pierwsza ) powinno zostać sprawdzone czy któryś z graczy ma możliwość wykonania jakiegoś ruchu. Jeśli nie to oznacza to koniec gry. Jeśli ruch ma ten gracz na którego przyszła kolej to czekamy na jego posunięcie, jeśli nie zmieniamy turę.

2) Kiedy gracz wskaże dane pole celem wykonania ruchu, należy sprawdzić czy ruch był legalny i wykonać lub zabronić wykonania takiego posunięcia.

3) Jeśli ruch okazał się legalny należy pozmieniać kolory odpowiednich pionków na planszy i zmienić turę na następną.

4) Na koniec sprawdzić czy nie był to czasem ostatni możliwy do wykonania ruch, który kończy tym samym rozgrywkę.

Teraz kiedy mamy już ogólne pojęcie o tym co pozostało do zrobienia, przystępujemy do pisania kodu. Zauważmy ze aby wykonać ruch gracz musi kliknąć w pole planszy. Dlatego właśnie przypadek „left_pressed” będzie idealnym miejscem na wstawienie skryptu, który wykonuje te wszystkie czynności. W tym miejscu powiem jeszcze, że w celu optymalizacji kodu należy za wszelką cenę unikać działów „step” oraz „draw”. Wykonują one zadeklarowane instrukcje w każdym cyklu zegarowym procesora ( często niepotrzebnie ) co znacznie obciąza procesor. Wracając do wstawiania pionkow... Tworzymy skrypt o nazwie „wstaw_piona” a tam wpisujemy odpowiedni kod :

if global.tura==0
{
    if sprawdz_ruch(x,y,global.tura)
    {
    global.tablica[x/60,y/60]=global.tura; 
    instance_create(x,y,pion);
    global.zmiany_tur=0;
    if sprawdz_czy_ruch(1)
    {
    global.tura=1;
    Tura.image_single=global.tura;
    } else sprawdz_czy_ruch(0)
    instance_destroy();
    }
}
else
{
    if sprawdz_ruch(x,y,global.tura)
    {
    global.tablica[x/60,y/60]=global.tura; 
    instance_create(x,y,pion);
    global.zmiany_tur=0;
    if sprawdz_czy_ruch (0)

    {

    global.tura=0;

    Tura.image_single=global.tura;

    } else sprawdz_czy_ruch(1)

    instance_destroy();

    }

}

Należy się wam trochę wyjaśnień. Zwłaszcza, że nazwy niektórych skryptów pojawiają się tu po raz pierwszy. Skrypt rozpoczyna działanie po kliknięciu w pole planszy. Pierwszą rzeczą jaką sprawdza jest aktualna tura. Dzieje się tak dlatego, że inne czynności zostaną wykonane dla pionków białych a inne dla czarnych. ( Można ten kod zoptymalizować ale poprawa jaką zyskamy w ten sposób będzie niczym w porównaniu z wysiłkiem jaki musielibyście włożyć w zrozumienie interpretacji tego rozwiązania. Jako, że artykuł przeznaczony jest dla osób średnio zaawansowanych pozostaniemy przy aktualnym rozwiązaniu. )

Po określeniu aktualnej tury, należy sprawdzić czy żądana operacja ( wstawienia pionka ) jest dozwolona. Będzie to robił oddzielny skrypt. Podzielenie problemu na mniejsze części jest jedną z zalecanych technik programowania. Złożony problem podzielony na łatwiejsze zadania staje się dużo łatwiejszy do rozwiązania, a kod napisany w ten sposób jest o wiele bardziej przejrzysty i czytelny.

Skrypt sprawdz_ruch zwracać będzie jedną z dwóch wartości ( true lub false ) dlatego dozwolony jest zapis :

if sprawdz_ruch(x,y,global.tura)

Jak widać do skryptu przekazujemy 3 parametry. Współrzędne pola dla którego należy sprawdzić ruch oraz turę czyli w istocie gracza który wykonuje ruch. Jest to minimalna ilość informacji konieczna do określenia poprawności ruchu. Dokładniej tym skryptem zajmiemy się za chwile.

Jeśli okaże się że żądany ruch jest wykonalny na miejscu pola zostanie stworzony pion a w odpowiednie miejsce w tablicy zostanie wpisana wartość definiująca jego kolor. Należy pamiętać o tym, że kolor pionka zależy jednoznacznie od liczby wpisanej w tablicy, dlatego zanim stworzymy obiekt należy aktualizować tablicę. 

global.tablica[x/60,y/60]=global.tura; 
instance_create(x,y,pion);

Pomińmy na chwilę zapis : global.zmiany_tur=0; wrócimy do niego za chwilę.

Tak oto wstawiliśmy pionka na dane pole. Spostrzeżeniem, które może nam zaoszczędzić wiele kłopotów jest fakt, że gra może się skończyć jedynie wówczas kiedy ktoś wstawi odpowiedni pionek. Pozwoli nam to sprawdzać warunki końca gry jedynie wtedy kiedy dany ruch został wykonany i co więcej był poprawny. Nie trzeba zatem robić tego za każdym razem w dziale „step” jak robi to większość mniej doświadczonych twórców gier.

if sprawdz_czy_ruch(1)
    {
    global.tura=1;
    Tura.image_single=global.tura;
    } else sprawdz_czy_ruch(0)

Następnie sprawdzamy, czy gdybyśmy zmienili turę to gracz który miałby wówczas ruch będzie miał do wykonania jakikolwiek ruch. Gdyby bowiem okazało się, że nie, to nie ma po co zmieniać tury. Wykonujemy skrypt sprawdz_czy_ruch w parametrze podając turę jaka będzie po zmianie i dopiero jeśli okaże się że w kolejnej turze będzie jakiś ruch do wykonania zmieniamy turę. Dodatkowo pamiętamy, że może wystąpić sytuacja, kiedy żaden z graczy nie może już wykonać ruchu, co oznaczałoby koniec gry. Dlatego jeśli okaże się, że po zmianie tury żaden ruch nie będzie dostępny, sprawdzamy czy znajdzie się jakiś ruch dla aktualnego gracza. Zliczamy przy okazji zmiany tur. Jeśli okaże się, że żaden z graczy nie będzie miał ruchu ( będą 2 zmiany tur bez wykonania ruchu ) będzie można uznać ze gra jest skonczona. Jeśli jednak ruch zostanie wykonany, należy wyzerować licznik zmian. Tym właśnie zajmowała się linijka, którą wcześniej pomineliśmy.

global.zmiany_tur=0;

Sprawdzamy czy ruch jest dozwolony...
Jak wspomniałem wcześniej, przed wstawieniem pionka, należy sprawdzić czy ruch jest wykonalny. Robimy to wywołując skrypt :

sprawdz_ruch(x,y,global.tura)

Zanim zaczniemy pisać taki skrypt, należy się zastanowić kiedy można uznać, że żądane posunięcie jest istotnie legalne. W przypadku naszej gry jest to stosunkowo proste, aczkolwiek kod zajmuje trochę miejsca i bez uprzedniego wyjaśnienia mógłby się nawet wydawać skomplikowany ;-)

Przede wszystkim wskazane pole musi być wolne. To chyba oczywiste, że nie można wskazać pola na którym jest już pionek przeciwnika a tym bardziej nasz własny. Tu właśnie przyda się tablica. Zamiast sprawdzać czy na danej pozycji jest już pionek, sprawdzimy jaka liczba jest wpisana w tablicy. Następnie należy przesuwając się w każdą stronę od miejsca w które klikneliśmy sprawdzać czy na pozycji „obok” znajduje się pionek przeciwnika, czyli ten, którego tury właśnie nie ma. Chciałbym to podkreślić, gdyż sprawdzenie czy pionek jest przeciwnika za pomocą dyrektywy „if” nie jest wcale takie proste. Można do tego użyć kodu

If argument2(czyli nasza global.tura) == 1

{

//wpisać kod sprawdzający czy pionek obok jest koloru 0

}

else

If argument2(czyli nasza global.tura) == 0

{

//wpisać kod sprawdzający czy pionek obok jest koloru 1

}

Takie rozwiązanie ( chociaż poprawne ) nastręcza wielu kombinacji i strasznie komplikuje kod. Tutaj po raz kolejny zyskujemy na tym, że do oznaczeń tury i pionków użyliśmy liczb 0 i 1. To czego nam trzeba to sprawny algorytm zmieniający liczbę 0 na 1 i odwrotnie. Wbrew pozorom taki algorytm wcale nie jest taki trudny. Zauważcie, że:

0*(-1)+1=1

1*(-1)+1=0

Co więcej Game Maker przyjmuje domyślnie, że jedynka oznacza prawdę a zero fałsz. Dzięki temu ten sam efekt uzyskamy używając negacji czyli operatora „!”. Możemy zatem wpisać kod sprawdzający czy pionek obok ma kolor „!nasz” ( czytaj nie nasz ).

Kiedy już okaże się, że pionek znajdujący się obok jest pionkiem przeciwnika, wystarczy „iść” w tym kierunku tak długo, aż napotkamy na pole puste ( co będzie oznaczać ze w tą stronę nie można wykonać ruchu ) lub na pionek naszego koloru ( co będzie oznaczać ze należy odmienić pionki które minęliśmy w międzyczasie sprawdzania. Do odwracania pionków będzie służył prosty skrypt zamien_piona(x,y) który za pomocą negacji zamienia aktualny kolor na przeciwny.

with(instance_position(argument0,argument1,pion))
{ 
image_single=image_single*(-1)+1;
global.tablica[argument0/60,argument1/60]=global.tablica[argument0/60,argument1/60]*(-1)+1;
}
Czynności te należy powtórzyć 8 razy dla wszystkich kierunków linii w jakich mogą ułożyć się pionki, uważając przy tym żeby nie „wyjść” poza obszar planszy. Cały skrypt będzie wyglądał następująco.

mozliwy=false;
if (argument0/60)<8
{
    if global.tablica[argument0/60+1,argument1/60]==(argument2*(-1))+1
    {
    xx=argument0/60+1
    do
        {
        xx+=1
        if global.tablica[xx,argument1/60]==argument2 
            {
            mozliwy=true; 
            for (i=xx-1;i>argument0/60;i-=1)
            zamien_piona(i*60,argument1);
            }
        }
        until (global.tablica[xx,argument1/60]=argument2||xx==9||global.tablica[xx,argument1/60]=-1)
    }
}
if (argument0/60)>3
{
    if global.tablica[argument0/60-1,argument1/60]==(argument2*(-1))+1
    {
    xx=argument0/60-1
    do
        {
        xx-=1
        if global.tablica[xx,argument1/60]==argument2 
            { 
            mozliwy=true;
            for (i=xx+1;i<argument0/60;i+=1)
            zamien_piona(i*60,argument1);
            }
        }
        until (global.tablica[xx,argument1/60]=argument2||xx==2||global.tablica[xx,argument1/60]=-1)
    }
}
if (argument1/60)<7
{
    if global.tablica[argument0/60,argument1/60+1]==(argument2*(-1))+1
    {
    yy=argument1/60+1
    do
        {
        yy+=1
        if global.tablica[argument0/60,yy]==argument2 
            { 
            mozliwy=true;
            for (i=yy-1;i>argument1/60;i-=1)
            zamien_piona(argument0,i*60);
            }
        }
        until (global.tablica[argument0/60,yy]=argument2||yy==8||global.tablica[argument0/60,yy]=-1)
    }
}
if (argument1/60)>2
{
    if global.tablica[argument0/60,argument1/60-1]==(argument2*(-1))+1
    {
    yy=argument1/60-1
    do
        {
        yy-=1
        if global.tablica[argument0/60,yy]==argument2 
            { 
            mozliwy=true;
            for (i=yy+1;i<argument1/60;i+=1)
            zamien_piona(argument0,i*60);
            }
        }
        until (global.tablica[argument0/60,yy]=argument2||yy==1||global.tablica[argument0/60,yy]=-1)
    }
}
if (argument0/60)<8 && argument1/60<7
{
    if global.tablica[argument0/60+1,argument1/60+1]==(argument2*(-1))+1
    {
    xx=argument0/60+1
    yy=argument1/60+1
        do
        {
        yy+=1
        xx+=1
        if global.tablica[xx,yy]==argument2 
            { 
            j=xx-1;
            mozliwy=true;
            for (i=yy-1;i>argument1/60;i-=1)
                {
                zamien_piona(j*60,i*60);
                j-=1;
                }
            }
        }
        until (global.tablica[xx,yy]=argument2||xx==9||yy==8||global.tablica[xx,yy]=-1)
    }
}
if (argument0/60)>3 && argument1/60>2
{
    if global.tablica[argument0/60-1,argument1/60-1]==(argument2*(-1))+1
    {
    xx=argument0/60-1
    yy=argument1/60-1
        do
        {
        yy-=1
        xx-=1
        if global.tablica[xx,yy]==argument2 
            { 
            j=xx+1;
            mozliwy=true;
            for (i=yy+1;i<argument1/60;i+=1)
                {
                zamien_piona(j*60,i*60);
                j+=1;
                }
            }
        }
        until (global.tablica[xx,yy]=argument2||yy==1||xx==2||global.tablica[xx,yy]=-1)
    }
}
if (argument0/60)>3 && argument1/60<7
{
    if global.tablica[argument0/60-1,argument1/60+1]==(argument2*(-1))+1
    {
    xx=argument0/60-1
    yy=argument1/60+1
        do
        {
        yy+=1
        xx-=1
        if global.tablica[xx,yy]==argument2 
            { 
            j=xx+1;
            mozliwy=true;
            for (i=yy-1;i>argument1/60;i-=1)
                {
                zamien_piona(j*60,i*60);
                j+=1;
                }
            }
        }
        until (global.tablica[xx,yy]=argument2||yy==8||xx==2||global.tablica[xx,yy]=-1)
    }
}
if (argument0/60)<8 && argument1/60>2
{
    if global.tablica[argument0/60+1,argument1/60-1]==(argument2*(-1))+1
    {
    xx=argument0/60+1
    yy=argument1/60-1
        do
        {
        yy-=1
        xx+=1
        if global.tablica[xx,yy]==argument2 
            { 
            j=xx-1;
            mozliwy=true;
            for (i=yy+1;i<argument1/60;i+=1)
                {
                zamien_piona(j*60,i*60);

                j-=1;

                }

            }

        }

        until (global.tablica[xx,yy]=argument2||yy==1||xx==9||global.tablica[xx,yy]=-1)

    }

}

if mozliwy return true else return false;

Kolejna rzecz jaką sprawdzaliśmy to czy po wykonaniu ruchu nie należy stwierdzić, że to już koniec gry. Do problemu można podejść w dwojaki sposób. Można by wymieniać przypadki, w których gra się zakończy ( nie ma już wolnych pól lub są same białe lub czarne pionki itd.. ) lub podejść do tego bardziej ogólnie. Zauważmy, że gra się skończy wtedy i tylko wtedy, jeśli żadna ze stron nie będzie miała już żadnych ruchów do wykonania. Takie stwierdzenie obejmuje także przypadek kiedy na planszy został już tylko jeden kolor lub nie ma już wolnych pól, gdyż wówczas również żadna ze stron nie będzie mogła wykonać ruchu. Jak więc sprawdzić czy został jeszcze do wykonania jakiś ruch ? Wykonujemy to bardzo podobnie jak w przypadku sprawdzania jednego pola, z tą róznicą że sprawdzamy warunek dla wszystkich wolnych pól, przerywając sprawdzanie kiedy napotkamy na dozwolony ruch lub koncząc grę w przeciwnym wypadku. Modyfikujemy skrypt sprawdz_ruch i otwrzymujemy :

for (k=2;k<10;k+=1)
    for (r=1;r<9;r+=1)
    {
    if global.tablica[k,r]=-1 
{    
if (k)<8
{
    if global.tablica[k+1,r]==(argument0*(-1))+1
    {
    xx=k+1
    do
        {
        xx+=1
        if global.tablica[xx,r]==argument0 
            {
            return true; 
            }
        }
        until (global.tablica[xx,r]=argument0||xx==9||global.tablica[xx,r]=-1)
    }
}
if (k)>3
{
    if global.tablica[k-1,r]==(argument0*(-1))+1
    {
    xx=k-1
    do
        {
        xx-=1
        if global.tablica[xx,r]==argument0 
            { 
            return true;
            }
        }
        until (global.tablica[xx,r]=argument0||xx==2||global.tablica[xx,r]=-1)
    }
}
if (r)<7
{
    if global.tablica[k,r+1]==(argument0*(-1))+1
    {
    yy=r+1
    do
        {
        yy+=1
        if global.tablica[k,yy]==argument0 
            { 
            return true;
            }
        }
        until (global.tablica[k,yy]=argument0||yy==8||global.tablica[k,yy]=-1)
    }
}
if (r)>2
{
    if global.tablica[k,r-1]==(argument0*(-1))+1
    {
    yy=r-1
    do
        {
        yy-=1
        if global.tablica[k,yy]==argument0 
            { 
            return true;
            }
        }
        until (global.tablica[k,yy]=argument0||yy==1||global.tablica[k,yy]=-1)
    }
}
if (k)<8 && r<7
{
    if global.tablica[k+1,r+1]==(argument0*(-1))+1
    {
    xx=k+1
    yy=r+1
        do
        {
        yy+=1
        xx+=1
        if global.tablica[xx,yy]==argument0 
            { 
            return true;
            }
        }
        until (global.tablica[xx,yy]=argument0||xx==9||yy==8||global.tablica[xx,yy]=-1)
    }
}
if (k)>3 && r>2
{
    if global.tablica[k-1,r-1]==(argument0*(-1))+1
    {
    xx=k-1
    yy=r-1
        do
        {
        yy-=1
        xx-=1
        if global.tablica[xx,yy]==argument0 
            { 
            return true;
            }
        }
        until (global.tablica[xx,yy]=argument0||yy==1||xx==2||global.tablica[xx,yy]=-1)
    }
}
if (k)>3 && r<7
{
    if global.tablica[k-1,r+1]==(argument0*(-1))+1
    {
    xx=k-1
    yy=r+1
        do
        {
        yy+=1
        xx-=1
        if global.tablica[xx,yy]==argument0 
            { 
            return true;
            }
        }
        until (global.tablica[xx,yy]=argument0||yy==8||xx==2||global.tablica[xx,yy]=-1)
    }
}
if (k)<8 && r>2
{
    if global.tablica[k+1,r-1]==(argument0*(-1))+1
    {
    xx=k+1
    yy=r-1
        do
        {
        yy-=1
        xx+=1
        if global.tablica[xx,yy]==argument0 
            { 
            return true;

            }

        }

        until (global.tablica[xx,yy]=argument0||yy==1||xx==9||global.tablica[xx,yy]=-1)

    }

}}}

global.zmiany_tur+=1;

if global.zmiany_tur=2 { Licz.alarm[0]=10; }

return false;

Kiedy okaże się, że gra jest skonczona uruchamiamy alarm[0] po czym wyświetlamy komunikat o zakończeniu rozgrywki.

Mamy zatem planszę, pionki i cały engine gry pozwalający na rozgrywkę dwu-osobową. Aby uatrakcyjnić partię dodamy na koniec obiekt który będzie wyświetlał aktualną turę oraz liczył punkty ( pionki ) obu graczy.

W tym celu stworzyłem sobie mniejszą kopię sprite’a pionka i wyświetlam odpowiednią klatkę sprite’a za pomocą kodu obiekt.image_single=global.tura

Myślę, że na tym etapie nie muszę już tego wyjaśniać. Do liczenia punktów stworzyłem obiekt Licz który wyświetla 2 sprit’y ( biała ikonka pionka oraz czarna ikonka pionka ) a pod spodem wyświetla ilość pionków wyliczoną za pomocą skryptu licz_pionki , który uruchamiam po każdym wykonanym ruchu.

Licz.bialych=0;
Licz.czarnych=0;
for (i=2;i<10;i+=1)
    for (j=1;j<9;j+=1)
    {
    if global.tablica[i,j]=0 Licz.czarnych+=1;
    if global.tablica[i,j]=1 Licz.bialych+=1;
    }

W ten oto sposób doszliśmy do końca artykułu i mamy gotową grę logiczną. Zdaję sobie sprawę, że wymaga ona jeszcze nieco ładniejszej grafiki, menu , tabeli wyników oraz chociażby gry z komputerem. Gra napisana została jednak w celach edukacyjnych i służy tylko jako przykład.
